PINCH_RESISTOR2

```
* dsauersanjose@aol.com  8/15/08
* www.idea2ic.com
*
*                XRpinch
*     Vtest     _________          Rir
*       _____|_/\  /\  /\___/\  /\  /\_
*      _|__       \/  \/      \/  \/  _|_
*     /_   \       :       Vir         ///
*    //  \  \      :
*    \    \_//     :        Rset (program R value)
*     \____/       :./\  /\  /\_
*       _|_            \/  \/  _|_
*       ///     VRset          ///
*    Vramp


Vramp    Vtest  0     PWL(   0     0.1    8    8.0)
XRpinch  Vtest  Vir   VRset  Rpinch
Rir      Vir    0     1
Rset     VRset  0     10k
.tran    100m   8     0       100m

.control
run
plot    vir   ylimit 0 .2m
dump
.endc


*      Rpinch layout
*      Attach Pin 2 to highest voltage (will zenor at 7volts)
*
*
*        ________________________________________________________
*        /                                           Nepi        /
*       /  ____________ /.....................\_______________  /
*      / /  ________   /                     /   ________   /  /
*     / / /____  1 /  /          2          /   /____  3 / /  /
*    / / //___/| /   /                     /   //___/| / /  /
*   / / / |___|//   /                     /   / |___|// /  /
* _/_/_/______/____/_____________________/_____/_______/_/__/___
*  | | | p+imp |   |        n+imp         |     | p+imp | |   |
*  |  \ \_____/    \___________________/  base  \_____/ /    |
*  |   \____________________________________________/       |  Psub
*  |                                                        |
*  |/  _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _  _ _ _ _ _ _ _\|
*   \  N BURIED LAYER                                      /
*   \ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ _ /
*        P-Substrate
*
*                                  .SUBCKT  Rpinch   INP   INN  VRS
*  |INP|____________                |VRS|
*  |___|    |   |   _|_             |___|
*           _|_/ |_ /   \             _|_      ==>   |INP|_|_/\  /\  /\_|INN|
*          / ^     |Bpinc|           / _ \           |___|    \/  \/    |___|
*  Dzenor/_\      _|     |           \/ \/                      _:_
*           |     | \___/            /\_/\                      |VRS|
*           |     |     |            \___/Isen                  |___|  Used to set Rvalue
*           |     |     |              _|_
*  |INN|____|_____|_____|              ///
*  |___|
*
.SUBCKT  Rpinch   INP    INN   VRS
Isen     VRS      0      -1u
Bpinc    INP      INN    I = .000001*(v(INP) - v(INN))/(v(VRS)*(1+1*(v(INP) - v(INN))))
Dzenor   INN      INP    Dzen
.ENDS    Rpinch

.MODEL    Dzen     D( RS=10 BV=7  )
```

.end

======Rpinch_Curver_Tracer_Plot============

The pinch resistors model may sometimes need to be extended to include the zenor behaviour.



***#1============WinSpiceVersion================**

```
PINCH_RESISTOR2
* dsauersanjose@aol.com  8/15/08
* www.idea2ic.com
*
*                XRpinch
*     Vtest      _______             Rir
*       _____|_/\  /\  /\___/\  /\  /\_
*     _|__      \/  \/        \/  \/   _|_
*    /_   \       :       Vir          ///
*   //\    \      :
*   \   \_//      :        Rset (program R value)
*    \____/       :./\  /\  /\_
*      _|_            \/  \/   _|_
*     ///     VRset           ///
*    Vramp
```

```
Vramp    Vtest  0    PWL(   0     0.1   8   8.0)
XRpinch  Vtest  Vir  VRset  Rpinch
Rir      Vir    0    1
Rset     VRset  0    10k
.tran    100m   8    0      100m

.control
run
plot    vir   ylimit 0 .2m
dump
.endc


*      Rpinch layout
*      Attach Pin 2 to highest voltage (will zenor at 7volts)
```




```
.SUBCKT  Rpinch   INP   INN  VRS
Isen     VRS      0     -1u
Bpinc    INP      INN   I = .000001*(v(INP) - v(INN))/(v(VRS)*(1+1*(v(INP) - v(INN))))
Dzenor   INN      INP   Dzen
.ENDS    Rpinch

.MODEL    Dzen    D( RS=10 BV=7 )

.end
```